#region license
// Copyright (c) 2007 Rodrigo B. de Oliveira (rbo@acm.org)
// All rights reserved.
// 
// Redistribution and use in source and binary forms, with or without modification,
// are permitted provided that the following conditions are met:
// 
//     * Redistributions of source code must retain the above copyright notice,
//     this list of conditions and the following disclaimer.
//     * Redistributions in binary form must reproduce the above copyright notice,
//     this list of conditions and the following disclaimer in the documentation
//     and/or other materials provided with the distribution.
//     * Neither the name of Rodrigo B. de Oliveira nor the names of its
//     contributors may be used to endorse or promote products derived from this
//     software without specific prior written permission.
// 
// THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS" AND
// ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE IMPLIED
// WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE ARE
// DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT OWNER OR CONTRIBUTORS BE LIABLE
// FOR ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR CONSEQUENTIAL
// DAMAGES (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF SUBSTITUTE GOODS OR
// SERVICES; LOSS OF USE, DATA, OR PROFITS; OR BUSINESS INTERRUPTION) HOWEVER
// CAUSED AND ON ANY THEORY OF LIABILITY, WHETHER IN CONTRACT, STRICT LIABILITY,
// OR TORT (INCLUDING NEGLIGENCE OR OTHERWISE) ARISING IN ANY WAY OUT OF THE USE OF
// THIS SOFTWARE, EVEN IF ADVISED OF THE POSSIBILITY OF SUCH DAMAGE.

# Taken and modified from ConcurrentBoo

# This file is part of Grawlix.
# 
# Grawlix is free software: you can redistribute it and/or modify
# it under the terms of the GNU General Public License as
# published by the Free Software Foundation, either version 3 of
# the License, or (at your option) any later version.
# 
# Grawlix is distributed in the hope that it will be useful,
# but WITHOUT ANY WARRANTY; without even the implied warranty of
# MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE.  See the
# GNU General Public License for more details.
# 
# You should have received a copy of the GNU General Public
# License along with Grawlix.  If not, see
# <http://www.gnu.org/licenses/>.
#endregion

"""Higher-order procedures upon concurrent structures."""

namespace ConcurrentBoo
import ConcurrentBoo.Primitive
import System.Collections

static class Functional:
"""Functional operations on collections."""

	callable MappingFunction(item) as object
	"""A function returning a value."""
	
	def Map(items as IEnumerable, operation as MappingFunction):
	"""
	Concurrently map a collection onto a function.
	Param items: A collection of values.
	Param operation: The function to invoke.
	Returns: An enumerator yielding the results.
	""" 
		for computation in MapComputation(items, operation):
			yield computation.Value
			
	def MapUnordered(items as IEnumerable, operation as MappingFunction):
	"""
	Map a function onto a collection of values.
	Param items: A collection of values.
	Param operation: The function to invoke.
	Returns: An enumerator yielding the results.
	Remarks: Mapped items are yielded as they become available.
	"""
		for computation in MapUnorderedComputation(items, operation):
			yield computation.Value
	
	def MapComputation(items as IEnumerable, operation as MappingFunction):
	"""
	Concurrently map a collection onto a function, returning Computations.
	Param items: A collection of values.
	Param operation: The function to invoke.
	Returns: An enumerator yielding Computation objects.
	"""
		computations = [Future({ operation(item) }) for item in items]
		for computation as Computation in computations:
			yield computation
	
	def MapUnorderedComputation(items as IEnumerable,
	                            operation as MappingFunction):
	"""
	Concurrently map a collection onto a function, unordered,
	returning Computations.
	Param items: A collection of values.
	Param operation: The function to invoke.
	Returns: An enumerator yielding Computation objects.
	Remarks: Mapped items are yielded as they become available.
	"""
		channel = ComputationChannel()
		
		total = 0
		for item in items:
			channel.Enqueue({ operation(item) })
			++total
		
		while total > 0:
			--total
			yield channel.Dequeue()
			
